G08ECF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G08ECF performs the triplets test on a sequence of observations from the interval [0,1].

2 Specification

```
 SUBROUTINE G08ECF(CL, N, X, MSIZE, NCOUNT, LDC, EX, CHI, DF, PROB,
1                  IFAIL)
 INTEGER           N, MSIZE, NCOUNT(LDC,LDC,MSIZE), LDC, IFAIL
 real              X(N), EX, CHI, DF, PROB
 CHARACTER*1       CL
```

3 Description

G08ECF computes the statistics for performing a triplets test which may be used to investigate deviations from randomness in a sequence of [0,1] observations.

An m by m matrix, C, of counts is formed as follows: the element c_{jkl} of C is the number of triplets (X(i), X($i+1$), X($i+2$)) for $i = 1, 4, 7, \ldots, n-2$, such that

$$\begin{aligned} \frac{j-1}{m} \leq &\quad \mathrm{X}(i) &< \frac{j}{m} \\ \frac{k-1}{m} \leq &\quad \mathrm{X}(i+1) &< \frac{k}{m} \\ \frac{l-1}{m} \leq &\quad \mathrm{X}(i+2) &< \frac{l}{m}. \end{aligned}$$

Note that all triplets formed are non-overlapping and are thus independent under the assumption of randomness.

Under the assumption that the sequence is random, the expected number of triplets for each class (i.e., each element of the count matrix) is the same, that is the triplets should be uniformly distributed over the unit cube $[0,1]^3$. Thus the expected number of triplets for each class is just the total number of triplets, $\sum_{j,k,l=1}^{m} c_{jkl}$, divided by the number of classes, m^3.

The χ^2 test statistic used to test the hypothesis of randomness is defined as:

$$X^2 = \sum_{j,k,l=1}^{m} \frac{(c_{jkl} - e)^2}{e}$$

where $e = \sum_{j,k,l=1}^{m} c_{jkl}/m^3$ = expected number of triplets in each class.

The use of the χ^2 distribution as an approximation to the exact distribution of the test statistic, X^2, improves as the expected value, e, increases.

G08ECF may be used in two different modes:

(i) a single call to G08ECF which computes all test statistics after counting the triplets.
(ii) multiple calls to G08ECF with the final test statistics only being computed in the last call.

The second mode is necessary if all the data do not fit into the memory. See parameter CL in Section 5 for details on how to invoke each mode.

4 References

[1] Morgan B J T (1984) *Elements of Simulation* Chapman and Hall

[2] Ripley B D (1987) *Stochastic Simulation* Wiley

[3] Dagpunar J (1988) *Principles of Random Variate Generation* Oxford University Press

5 Parameters

1: CL — CHARACTER*1 *Input*

On entry: indicates the type of call to G08ECF,

If CL = 'S', this is the one and only call to G08ECF (single call mode). All data are to be input at once. All test statistics are computed after counting of the triplets is complete.

If CL = 'F', this is the first call to the routine. All initializations are carried out and the counting of triplets begins. The final test statistics are not computed since further calls will be made to G08ECF.

If CL = 'I', this is an intermediate call during which counts of the triplets are updated. The final test statistics are not computed since further calls will be made to G08ECF.

If CL = 'L', this is the last call to G08ECF. The test statistics are computed after the final counting of the triplets is complete.

Constraint: CL = 'S', 'F', 'I' or 'L'.

2: N — INTEGER *Input*

On entry: the number of observations, n.

Constraints: if CL = 'S', then N ≥ 3, otherwise N ≥ 1.

3: X(N) — ***real*** array *Input*

On entry: the sequence of observations.

4: MSIZE — INTEGER *Input*

On entry: the size of the count matrix to be formed, m. MSIZE must not be changed between calls to G08ECF.

Constraint: MSIZE ≥ 2.

5: NCOUNT(LDC,LDC,MSIZE) — INTEGER array *Input/Output*

On entry: if CL = 'S' or 'F', NCOUNT need not be set.

If CL = 'I' or 'L', NCOUNT must contain the values returned by the previous call to G08ECF.

On exit: NCOUNT is an MSIZE by MSIZE by MSIZE matrix containing the counts of the number of triplets, c_{jkl}, for $j, k, l = 1, 2, \ldots, m$.

6: LDC — INTEGER *Input*

On entry: the first and second dimensions of the array NCOUNT as declared in the (sub)program from which G08ECF is called.

Constraint: LDC $\geq$ MSIZE.

7: EX — ***real*** *Output*

On exit: if CL = 'S' or 'L', (i.e., if it is a final exit) then EX contains the expected number of counts for each element of the count matrix.

Otherwise EX is not set.

8: CHI — ***real*** *Output*

On exit: if CL = 'S' or 'L', (i.e., if it is a final exit) then CHI contains the χ^2 test statistic, X^2 for testing the null hypothesis of randomness.

Otherwise CHI is not set.

9: DF — ***real*** *Output*

On exit: if CL = 'S' or 'L' (i.e., if it is a final exit) then DF contains the degrees of freedom for the χ^2 statistic.

Otherwise DF is not set.

10: PROB — ***real*** *Output*

On exit: if CL = 'S' or 'L' (i.e., if it is a final exit) then PROB contains the upper tail probability associated with the χ^2 test statistic, i.e., the significance level.

Otherwise PROB is not set.

11: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error or gives a warning (see Section 6).

For this routine, because the values of output parameters may be useful even if IFAIL $\neq 0$ on exit, users are recommended to set IFAIL to −1 before entry. **It is then essential to test the value of IFAIL on exit**.

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings specified by the routine:

IFAIL = 1

On entry, CL $\neq$ 'S', 'F', 'I' or 'L'.

IFAIL = 2

On entry, N < 1,
or CL = 'S' and N < 3.

IFAIL = 3

On entry, MSIZE ≤ 1.

IFAIL = 4

On entry, LDC $<$ MSIZE.

IFAIL = 5

On entry, X(i) < 0.0,
or X(i) > 1.0, for some $i = 1, 2, \ldots, n$.

IFAIL = 6

No triplets were found because less than 3 observations were provided in total.

IFAIL = 7

The expected value for the counts in each element of the count matrix is less than or equal to 5.0. This implies that the χ^2 distribution may not be a very good approximation to the distribution of the test statistic.

7 Accuracy

The computations are believed to be stable. The computations of PROB given the values of CHI and DF will obtain a relative accuracy of 5 significant figures for most cases.

8 Further Comments

If the call to G08ECF is an initial call or intermediate call with further calls to follow then any unused observations are saved for use at the beginning of the new sequence provided in the following call. Clearly any observations left over from an only or final call to G08ECF are ignored.

The time taken by the routine increases with the number of observations n, and depends to some extent whether the call is an only, first, intermediate or last call.

9 Example

The following program performs the pairs test on 10000 pseudo-random numbers from a uniform distribution $U(0,1)$ generated by G05CAF. G08ECF is called 10 times with 1000 observations on each call. The triplets are tallied into a 5 by 5 by 5 matrix.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G08ECF Example Program Text
*     Mark 14 Release.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          N, MSIZE, LDC
      PARAMETER        (N=1000,MSIZE=5,LDC=5)
*     .. Local Scalars ..
      real             CHI, DF, EX, P
      INTEGER          I, IFAIL, J, K
      CHARACTER*1      CL
*     .. Local Arrays ..
      real             X(N)
      INTEGER          NCOUNT(LDC,LDC,MSIZE)
*     .. External Subroutines ..
      EXTERNAL         G05CBF, G05FAF, G08ECF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G08ECF Example Program Results'
      CALL G05CBF(0)
      DO 20 I = 1, 10
         IF (I.EQ.1) THEN
            CL = 'First'
         ELSE IF (I.EQ.10) THEN
            CL = 'Last'
         ELSE
            CL = 'Intermediate'
         END IF
         CALL G05FAF(0.0e0,1.0e0,N,X)
         IFAIL = -1
*
         CALL G08ECF(CL,N,X,MSIZE,NCOUNT,LDC,EX,CHI,DF,P,IFAIL)
*
         IF (CL.NE.'L' .AND. CL.NE.'l' .AND. IFAIL.NE.0) GO TO 80
```

```
*
   20 CONTINUE
*
      IF (IFAIL.EQ.0 .OR. IFAIL.EQ.7) THEN
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Count matrix'
         DO 60 I = 1, MSIZE
            WRITE (NOUT,*)
            WRITE (NOUT,99999) 'I = ', I
            WRITE (NOUT,*) '        1      2      3      4      5'
            WRITE (NOUT,*)
            DO 40 J = 1, MSIZE
               WRITE (NOUT,99998) J, (NCOUNT(I,J,K),K=1,MSIZE)
   40       CONTINUE
   60    CONTINUE
         WRITE (NOUT,*)
         WRITE (NOUT,99997) 'Expected value = ', EX
         WRITE (NOUT,99996) 'CHISQ          = ', CHI
         WRITE (NOUT,99997) 'DF             = ', DF
         WRITE (NOUT,99996) 'Prob           = ', P
         IF (IFAIL.EQ.7) WRITE (NOUT,*)
     +   ' ** Note : the chi square approximation may not be very good.'
       END IF
   80 STOP
*
99999 FORMAT (1X,A,I2)
99998 FORMAT (1X,I2,5I7)
99997 FORMAT (1X,A,F8.2)
99996 FORMAT (1X,A,F10.4)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
G08ECF Example Program Results

Count matrix

I =  1
        1      2      3      4      5

 1     21     26     28     35     20
 2     30     27     23     32     45
 3     32     28     27     22     19
 4     36     33     31     33     28
 5     24     27     31     25     22

I =  2
        1      2      3      4      5

 1     26     29     29     26     26
 2     33     21     27     23     21
 3     26     26     29     24     19
 4     41     29     27     24     27
 5     23     36     31     23     33
```

```
I =  3
         1       2       3       4       5

 1      22      25      26      24      24
 2      21      23      33      33      29
 3      29      22      24      24      28
 4      24      29      18      30      25
 5      18      25      31      24      33

I =  4
         1       2       3       4       5

 1      24      29      31      26      21
 2      19      21      30      24      21
 3      27      20      23      22      33
 4      23      27      31      30      18
 5      27      36      24      29      22

I =  5
         1       2       3       4       5

 1      30      31      26      27      44
 2      33      34      23      21      28
 3      19      24      24      28      29
 4      26      22      16      26      36
 5      19      37      19      28      17

Expected value =    26.66
CHISQ          =   135.0093
DF             =   124.00
Prob           =     0.2353
```